

Paper 192-2010

An In-house SAS® Procedure to Balance Time Series: The TSRACING Procedure

Joana Bérubé, Statistics Canada, Ottawa, Canada

Susie Fortier, Statistics Canada, Ottawa, Canada

ABSTRACT

For various reasons, systems of time series might fail to satisfy cross-sectional aggregation constraints. To restore coherence in such systems (and as part of its Forillon project), Statistics Canada developed an in-house SAS® procedure called PROC TSRACING. This paper introduces the new procedure and describes its main features. The methodology implemented is illustrated with simple examples.

INTRODUCTION

Situations requiring raking are frequent in statistical agencies and are discussed in [1] and [2]. For various reasons, such as seasonal adjustment or the use of multiple sources of data, systems of time series may fail to satisfy cross-sectional aggregation constraints. The goal of raking is to restore these constraints. Optionally, temporal constraints can also be preserved.

A simple example of one-dimensional raking is to prorate regional estimates to ensure they add up to a known national total. A similar example of one-dimensional raking *with temporal constraints* would be to reconcile a year of quarterly regional estimates. For each quarter, regional estimates need to add up to the national total. Also, for each region, the annual total needs to be preserved.

An example of a situation requiring two-dimensional raking would be a system of time series classified by region and industry subgroups that are directly seasonally adjusted (SA). For each industry, the sum of the SA regions must add up to the directly SA national total. For each region, the sum of the SA industries must add up to the directly SA industrial total. This example can also be extended to one with temporal constraints when annual totals must be preserved.

To solve incoherence in a system of time series and to implement a raking solution, Statistics Canada developed an in-house raking SAS® procedure called PROC TSRACING using SAS/TOOLKIT® (see [3]). It is part of Statistics Canada's Forillon project. This project also includes an in-house benchmarking SAS® procedure called PROC BENCHMARKING (see [4]).

If external annual benchmarks are available, a univariate benchmarking method such as Forillon's PROC BENCHMARKING should be used to pre-adjust the annual levels (see [1] and [4]). The TSRACING procedure can preserve the annual totals. To do so, the procedure assumes that the series have been already benchmarked to their annual totals before the raking process begins.

This paper is organized as follows. First, the theory behind the procedure is introduced, followed by the syntax of the procedure and its main features. Illustrative examples are then set out. Finally, a particular example in which a repetitive use of the procedure is needed is provided.

THEORY BEHIND THE TSRACING PROCEDURE

The raking method implemented in the procedure is a specific, simple variant of the general regression-based raking model by Dagum and Cholette [5]. The theoretical details are provided in [1]. The model is as follows:

$$x = \theta + e, \quad e \sim (0, V_e)$$

$$g = G\theta + \varepsilon, \quad \varepsilon \sim (0, V_\varepsilon),$$

where

- x is a vector of the time series to be raked.
- θ is the raked/reconciled values of x .
- V_e is the variance matrix of e . $V_e = \text{diag}(c_x * x)$, where c_x is the vector of the alterability coefficients associated with x .
- G is the design matrix of additivity rules.
- g is the vector of constraints (marginal total values).

- V_ε is the variance matrix of ε . $V_\varepsilon = \text{diag}(c_g * g)$, where c_g is the vector of alterability coefficients associated with g .

The generalized least squared solution of the model in matrix form,

$$\begin{bmatrix} x \\ g \end{bmatrix} = \begin{bmatrix} I \\ G \end{bmatrix} \theta + \begin{bmatrix} e \\ \varepsilon \end{bmatrix}$$

is

$$\hat{\theta} = x + V_\varepsilon G' (G V_\varepsilon G' + V_\varepsilon)^{-1} (g - Gx)$$

The proofs can be found in Dagum and Cholette [5].

Often, the matrix $(G V_\varepsilon G' + V_\varepsilon)$ has not full rank. The Moore-Penrose generalized matrix inversion is then used in the above equation. This generalized inverse was chosen for its properties: when there are inconsistencies in the data, the generalized inverse will give a least squared solution. It also provides a unique solution [6].

The alterability coefficients, c_x and c_g in the above model, are numbers used to modify the relative size of the discrepancies allocated to a data point. The allocation is usually proportional to the data value. This allocation is achieved with equal alterability coefficients. (Almost) no discrepancies will be allocated to the data point if the coefficient is (almost) 0. More relative discrepancies can be allocated to some components if the coefficients are not all the same.

When its associated alterability coefficient is 0, a constraint or total is referred to as *binding*. In that case, all the discrepancies between the desired total and the sum of the components are allocated to the components. Otherwise, some of the discrepancies are allocated to the total itself (the total and the components are allowed to change).

If the total and most or all of component series have alterability coefficients of 0, then there could be no solution to the raking problem. By default, the alterability coefficients are equal to 1 for component series and 0 for totals.

In the model perspective, alterability coefficients could be used to modify the variance of the error associated with an observation or a constraint, so that they are more or less affected by the raking process.

THE TSRAKING PROCEDURE

The procedure allows one-dimensional and two-dimensional raking. If required, the annual total may be preserved without undue impact on the month-to-month movement.

PROCEDURE SYNTAX

```
PROC TSRAKING <option(s)>;
  ID variable(s);
RUN;
```

PROC TSRAKING STATEMENT

Use this option	To do this
METADATA=	Specify the metadata data set.
DATA=	Specify the input data set.
ALTERABILITY=	Specify the input alterability coefficients data set.
OUT=	Specify the output data set.
ALTERSERIES=	Specify the default alterability coefficient for the series.
ALERTOTAL1=	Specify the default alterability coefficient for total1.
ALERTOTAL2=	Specify the default alterability coefficient for total2.
ALTERANNUAL=	Specify the default alterability coefficient for the annual total.
TOLERANCEVALUE=	Specify a tolerance value for the binding totals, in absolute value.
TOLERANCEPERCENT=	Specify a tolerance value for the binding totals, in percentage.

OPTIONS

METADATA=SAS-data-set

It specifies the metadata SAS data set that describes the additivity rules used by the procedure. It is mandatory. Two character variables must be in this data set: SERIES and TOTAL1. Two variables are optional: TOTAL2 (character) and ALTERANNUAL (numeric).

The ALTERANNUAL variable could be used to specify the default alterability coefficient for the annual total for each series.

DATA=SAS-data-set

It specifies the SAS data set that contains the data. It is mandatory.

ALTERABILITY=SAS-data-set

It specifies the input alterability coefficients SAS data set. It is optional. If this data set is specified, the values of these alterability coefficients will override the values specified using the ALTERxxx options (see below).

OUT=SAS-data-set

It specifies the output SAS data set. It is optional. If it is not specified, PROC TSRAKING will create it by using the DATA n naming convention. If you want the OUT= data set to be permanent, specify a two-level name.

ALTERSERIES=positive real number

It specifies the default alterability coefficient for the SERIES variable. This number is optional and has a default of 1.0. It will be used if no other alterability coefficients have been specified for the SERIES variables (using the ALTERABILITY= SAS data set).

ALERTOTAL1=positive real number

It specifies the default alterability coefficient for the TOTAL1 variable. This number is optional and has a default of 0.0. It will be used if no other alterability coefficients have been specified for the TOTAL1 variables (using the ALTERABILITY= SAS data set).

ALERTOTAL2=positive real number

It specifies the default alterability coefficient for the TOTAL2 variable. This number is optional and has a default of 0.0. It will be used if no other alterability coefficients have been specified for the TOTAL2 variables (using the ALTERABILITY= SAS data set).

ALTERANNUAL=positive real number

It specifies the default alterability coefficient for the annual total. This number is optional and has a default of 0.0. It will be used if no values have been specified for the annual total (using the variable ALTERANNUAL of the METADATA= SAS data set).

If only this option is used, all of the series will have the same alterability coefficient for their annual totals.

The variable ALTERANNUAL in the METADATA= SAS data set should be used to state alterability coefficients for the annual total specific to each series.

TOLV=TOLERANCEVALUE=positive real number (including 0)

In the case where there are binding totals, this absolute value is the maximum difference permitted between those totals and the ones calculated by the procedure. This number is optional. If this option is not specified, the default value is 1E-3.

This threshold will be used to determine whether the raking process succeeded in restoring the additivity rules.

TOLP=TOLERANCEPERCENT= *real number (between 0.0 and 1.0)*

In the case where there are binding totals, this value is the maximum difference permitted between those totals and the ones calculated by the procedure, in percentage. This number is optional. There is no the default value if this option is not specified.

This threshold will be used to determine whether the raking process succeeded in restoring the additivity rules. It has to be specified in decimals: TOLERANCEPERCENT=0.05 means a threshold of 5%.

If both TOLERANCEVALUE= and TOLERANCEPERCENT= are specified, TOLERANCEVALUE= will be used. If neither of them is specified, then the default value of TOLERANCEVALUE= will be used.

ID STATEMENT

Use the ID statement to name the variables in the DATA= SAS data set that are not involved in the raking rules but are to be transferred from the input data set to the output data set OUT=. Variables listed in the ID statement must be contained in the DATA= SAS data set.

DETAILS

Observations with a missing value for any variable in the DATA= or the ALTERABILITY= SAS data sets will stop the procedure.

In the current version of the procedure, observations with a negative value for any variable in the ALTERABILITY= SAS data set will stop the procedure.

Each additivity rule of a system of time series must be defined separately. There must be a METADATA= SAS data set for each additivity rule. The procedure will have to be independently called for each rule, for each period. A period can be an entire year if the annual totals must be preserved (12 months, 4 quarters, etc.). It could also be a period (week, month, quarter, etc) for which there is no temporal constraint. Presently, the procedure must be run separately for each period that needs to be raked.

The procedure can preserve the annual totals. This is done via the DATA= SAS data set. If there is more than one line in the DATA= SAS data set, the procedure will assume that the raking must be done for each line while preserving the sum of each column. For example, if the data are monthly, there will be twelve lines in the DATA= SAS data set. For each series, the procedure will preserve the annual total. For illustrations, see examples 2 and 4 below.

If only one line is specified in the DATA= SAS data set, the procedure will assume that the raking is required for this period only. The period could be a day, a week, a month, a quarter, a year, etc. For illustrations, see examples 1, 3 and 5 below.

FUTURE DEVELOPMENTS

Future version of the procedure will include a BY statement for batch processing.

Simpler processing based on time id is also being considered. If there are temporal constraints, the procedure would first apply a raking rule for each complete year of data and then apply it period by period for the yet *to be completed* year.

EXAMPLE 1: ONE-DIMENSIONAL RAKING

CONTEXT

Suppose we have estimates of car sales (25 cars) and van sales (5 vans) for a given period. The total vehicle sales, called the control total, is known (40 vehicles). The sum of car and van sales is 30, not 40. Raking will be used to restore the coherence.

SAS PROCEDURE

```
DATA MyMetadata1;
LENGTH series $13 total1 $13;
series='Cars'; total1='Vehicles_Tot'; total2=''; OUTPUT;
series='Vans'; total1='Vehicles_Tot'; total2=''; OUTPUT;
RUN;
```

```

DATA MyData1;
INPUT Cars Vans Vehicles_Tot;
CARDS;
25 5 40
;
RUN;

PROC TSKRANK METADATA=MyMetadatal DATA=MyData1 OUT=raked_series1;
RUN;

```

RESULTS

The raked time series are Cars=33.33 and Vans=6.67 (results have been manually rounded). The sum of car and van sales is now equal to the total vehicle sales.

EXAMPLE 2: ONE-DIMENSIONAL RAKING WITH KNOWN ANNUAL TOTALS

CONTEXT

Suppose we have quarterly estimates of car sales for the East, Centre and West regions for a given year. The total car sales, called control totals, are known and annual totals must be preserved. The sum of the components is not equal to the control totals. For example, for the first quarter, the sum of car sales for the East, Centre and West is equal to 39. The true value is 40. The inconsistencies will all be resolved via raking. However, note that the sum of the control totals is equal to the sum of the annual totals, which is 142. The data are as follows:

	East	Centre	West	Sum	Control Total
Quarter 1	12	14	13	39	40
Quarter 2	10	9	15	34	25
Quarter 3	12	8	17	37	40
Quarter 4	9	9	14	32	37
Annual Total	43	40	59		

Suppose that the East car sales should not change for Quarter 2. Its associated alterability coefficient is set to 0. The control totals are also binding. For illustrative purposes, the vectors and matrices used for internal computation are shown below.

The vector containing the time series to be raked is

$$x = (12 \ 10 \ 12 \ 9 \ 14 \ 9 \ 8 \ 9 \ 13 \ 15 \ 17 \ 14)'$$

The vector of alterability coefficients associated with x is

$$c_x = (1 \ 0 \ 1 \ 1 \ 1 \ 1 \ 1 \ 1 \ 1 \ 1 \ 1 \ 1)'$$

The variance matrix of e is the element-wise product of the two previous vectors

$$V_e = \text{diag}(c_x * x) = \text{diag}(12, 0, 12, 9, 14, 9, 8, 9, 13, 15, 17, 14)$$

The design matrix of additivity rules is

$$G = \begin{pmatrix} \mathbf{I}_4 & \mathbf{I}_4 & \mathbf{I}_4 \\ \mathbf{1}_{1 \times 4} & \mathbf{0}_{1 \times 4} & \mathbf{0}_{1 \times 4} \\ \mathbf{0}_{1 \times 4} & \mathbf{1}_{1 \times 4} & \mathbf{0}_{1 \times 4} \\ \mathbf{0}_{1 \times 4} & \mathbf{0}_{1 \times 4} & \mathbf{1}_{1 \times 4} \end{pmatrix}$$

The identity matrices represent the contemporaneous constraints. The bottom three lines of the block matrix G will preserve the annual totals.

The marginal total values are

$$g = (40 \ 25 \ 40 \ 37 \ 43 \ 40 \ 59)'$$

The first four numbers are the totals associated with the contemporaneous constraints. The last three are the annual totals that must be preserved.

Both contemporaneous and annual totals are binding, so their associated alterability coefficient vector is

$$c_g = (0 \ 0 \ 0 \ 0 \ 0 \ 0 \ 0)'$$

The variance matrix of ε is the element-wise product of the two previous vectors

$$V_\varepsilon = \text{diag}(c_g * g) = \text{diag}(0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0)$$

SAS PROCEDURE

The TSRAKING procedure will be used to restore coherence in the system of time series. Preliminary steps must be completed before using it. The additivity rule must be described in a METADATA SAS data set. The mandatory variable SERIES contains the components, here the car sales for each region. The mandatory variable TOTAL1 contains the control total, here the total car sales. The variable TOTAL2 is empty because the raking rule is one-dimensional.

```
DATA MyMetadata2;
LENGTH series $10 total1 $10 total2 $10;
series='Cars_East'; total1='Cars_Total'; total2=''; OUTPUT;
series='Cars_Cent'; total1='Cars_Total'; total2=''; OUTPUT;
series='Cars_West'; total1='Cars_Total'; total2=''; OUTPUT;
RUN;
```

In the DATA SAS data set shown below, the first three columns are the series to be raked and the last one is the totals. Each line is a quarter.

```
DATA MyData2;
INPUT Cars_East Cars_Cent Cars_West Cars_Total;
CARDS;
12 14 13 40
10 9 15 25
12 8 17 40
9 9 14 37
;
RUN;
```

The ALTERABILITY SAS data set shown below contains all of the alterability coefficients. The first three columns are associated with the series to be raked and the last column is for the contemporaneous totals. The contemporaneous constraints are binding, so the last column is 0. Each line is a quarter. The 0 in the first column and second line means that the value for the Quarter 2 of the East-cars series must not change.

```
DATA MyAlter_Coeff2;
INPUT Cars_East Cars_Cent Cars_West Cars_Total;
CARDS;
1 1 1 0
0 1 1 0
1 1 1 0
1 1 1 0
;
RUN;
```

No explicit alterability coefficients are specified for the annual totals. The default value 0.0 is used. The annual totals are then binding.

```
PROC TSRAKING METADATA=MyMetadata2 DATA=MyData2
ALTERABILITY=MyAlter_Coeff2 OUT=raked_series2;
RUN;
```

RESULTS

The raked time series are as follows (results have been formatted into a table display and manually rounded):

	East	Centre	West	Control Total
Quarter 1	11.34	14.82	13.84	40
Quarter 2	10	5.59	9.41	25
Quarter 3	12.02	8.92	19.06	40
Quarter 4	9.64	10.67	16.69	37
Annual Total	43	40	59	

The binding contemporaneous totals are met for each quarter. The annual totals are preserved.

East car sales did not change for Quarter 2, as specified via the MyAlter_Coeff2 SAS data set.

EXAMPLE 3: TWO-DIMENSIONAL RAKING

CONTEXT

Suppose we have estimates of car and van sales for the East, Centre and West regions for the current quarter. The total car, van, East, Centre and West sales, called control totals, are known. There is no need to preserve the annual total, because only the current quarter is available. This is a two-dimensional situation. The first dimension is the type of vehicle: car or van. The second dimension is the geographical region: East, Centre or West. The sums of the components are not equal to the control totals. However, note that the sum of the regional control totals is equal to the sum of the vehicle control totals, which is 93. The data are as follows:

	East	Centre	West	Sum	Control Total
Cars	12	14	13	39	40
Vans	20	20	24	64	53
Sum	32	34	37		
Control Total	30	31	32		

There is no need to specify the alterability coefficients in this case. The totals are binding by default, whereas the components are not.

The matrices behind the procedure are the following:

$$\begin{aligned}
 x &= (12 \ 14 \ 13 \ 20 \ 20 \ 24)' \\
 c_x &= (1 \ 1 \ 1 \ 1 \ 1 \ 1)' \\
 V_e &= \text{diag}(12, 14, 13, 20, 20, 24) \\
 g &= (40 \ 53 \ 30 \ 31 \ 32)' \\
 c_g &= (0 \ 0 \ 0 \ 0 \ 0)' \\
 V_\varepsilon &= \text{diag}(0, 0, 0, 0, 0)
 \end{aligned}
 \quad
 G = \begin{pmatrix}
 \mathbf{1}_{1 \times 3} & \mathbf{0}_{1 \times 3} \\
 \mathbf{0}_{1 \times 3} & \mathbf{1}_{1 \times 3} \\
 \mathbf{I}_3 & \mathbf{I}_3
 \end{pmatrix}$$

The first two lines of the block matrix G represent the first dimension, which is the type of vehicle. The identity matrices represent the second dimension, which is the geographical dimension.

SAS PROCEDURE

```

DATA MyMetadata3;
LENGTH series $32 total1 $32 total2 $32;
series='Cars_East'; total1='Cars_Tot'; total2='East_Tot'; OUTPUT;
series='Cars_Cent'; total1='Cars_Tot'; total2='Cent_Tot'; OUTPUT;
series='Cars_West'; total1='Cars_Tot'; total2='West_Tot'; OUTPUT;
series='Vans_East'; total1='Vans_Tot'; total2='East_Tot'; OUTPUT;
series='Vans_Cent'; total1='Vans_Tot'; total2='Cent_Tot'; OUTPUT;
series='Vans_West'; total1='Vans_Tot'; total2='West_Tot'; OUTPUT;
RUN;

```

```

DATA MyData3;
INPUT Cars_East Cars_Cent Cars_West Vans_East Vans_Cent Vans_West Cars_Tot Vans_Tot
      East_Tot Cent_Tot West_Tot;
CARDS;
12 14 13 20 20 24 40 53 30 31 32
;
RUN;

PROC TSRAKING METADATA=MyMetadata3 DATA=MyData3 OUT=raked_series3;
RUN;

```

RESULTS

The raked time series are as follows (results have been formatted into a table display and manually rounded):

	East	Centre	West	Control Total
Cars	12.72	14.38	12.9	40
Vans	17.28	16.62	19.1	53
Control Total	30	31	32	

All of the contemporaneous constraints are respected for both dimensions.

EXAMPLE 4: TWO-DIMENSIONAL RAKING WITH KNOWN ANNUAL TOTALS

Suppose we have estimates of car and van sales for the East, Centre and West regions for a certain year. Data are quarterly. Suppose also that the annual totals must be preserved. This is a two-dimensional situation with the same raking rule as the previous example. As a result, the METADATA SAS data set and the procedure call would be the same as those in example 3. The DATA SAS data set would be the following:

```

DATA MyData4;
INPUT Cars_East Cars_Cent Cars_West Vans_East Vans_Cent Vans_West Cars_Tot Vans_Tot
      East_Tot Cent_Tot West_Tot;
CARDS;
12 14 13 19 19 24 38 64 30 34 38
14 17 13 18 22 21 35 53 28 31 29
10 13 11 17 18 19 45 59 33 38 33
18 20 16 23 25 28 53 77 40 45 45
;
RUN;

```

The raked time series are as follows (results have been formatted into a table display and manually rounded):

QUARTER 1	Cent	East	West	Control Total
Cars	13.93	11.24	12.83	38
Vans	20.07	18.76	25.17	64
Control Total	34	30	38	

QUARTER 3	Cent	East	West	Control Total
Cars	17.64	13.7	13.66	45
Vans	20.36	19.3	19.34	59
Control Total	38	33	33	

QUARTER 2	Cent	East	West	Control Total
Cars	12.79	11.73	10.48	35
Vans	18.21	16.27	18.52	53
Control Total	31	28	29	

QUARTER 4	Cent	East	West	Control Total
Cars	19.64	17.33	16.03	53
Vans	25.36	22.67	28.97	77
Control Total	45	40	45	

These four tables represent the two-dimensional raking rules for each one of the four periods. All the control totals are respected.

Finally, the temporal constraints have been preserved, as shown in the following table:

	Cars_Cent	Cars_East	Cars_West	Vans_Cent	Vans_East	Vans_West
Quarter 1	13.93	11.24	12.83	20.07	18.76	25.17
Quarter 2	12.79	11.73	10.48	18.21	16.27	18.52
Quarter 3	17.64	13.7	13.66	20.36	19.3	19.34
Quarter 4	19.64	17.33	16.03	25.36	22.67	28.97
Annual Total	64	54	53	84	77	92

EXAMPLE 5: MULTIPLE CALLS OF THE PROCEDURE

Suppose we have population size estimates for all five states in a given country. The total country population, called the control total, is known from another source. The estimates come from a survey conducted every 2 years. Suppose also that the data from the last 10 cycles are available. As the sum of the states differs from the country total, raking is needed. This is one-dimensional raking with no temporal constraint.

The raking rule is the same for each cycle. Therefore, there is only one METADATA SAS data set. The sum of states should be equal to the country total.

```
DATA MyMetadata5;
LENGTH series $32 total1 $32 total2 $32;
series='State1'; total1='Country_Tot'; total2=''; OUTPUT;
series='State2'; total1='Country_Tot'; total2=''; OUTPUT;
series='State3'; total1='Country_Tot'; total2=''; OUTPUT;
series='State4'; total1='Country_Tot'; total2=''; OUTPUT;
series='State5'; total1='Country_Tot'; total2=''; OUTPUT;
RUN;
```

The following is an intermediary data set. It includes all 10 cycles. Each line is a cycle, identified by the first column.

```
DATA Data_All;
INPUT Cycle State1 State2 State3 State4 State5 Country_Tot;
CARDS;
1 100 400 350 250 600 2000
2 200 450 500 300 750 2500
... ..
10 650 1300 1200 1050 2100 6500
;
RUN;
```

The following macro loops on each cycle. The first step is to subset the Data_All SAS data set so that it contains only one cycle at a time. Afterward, the TSRAKING procedure is used to reconcile the data. At the end, there will be ten outputted SAS data sets named raked_series1 to raked_series10.

```
%MACRO Do_Raking();
%DO i=1 %TO 10;
DATA MyData5;
SET Data_All (WHERE=(Cycle=&i));
RUN;

PROC TSRAKING METADATA=MyMetadata5 DATA=MyData5 OUT=raked_series&i;
RUN;
%END;
%MEND Do_Raking;
%Do_Raking()
```

In a future version of the procedure, a BY statement will be available. This will allow a simpler processing of the same rule on different observations. For example, the macro above would be replaced by the following procedure call:

```
PROC TSRAKING METADATA=MyMetadata5 DATA=Data_All OUT=raked_seriesAll;
BY Cycle;
RUN;
```

CONCLUSION

PROC TSRAKING is a simple way to restore coherence in a system of time series. It allows one- or two-dimensional reconciliation of contemporaneous constraints. It can also preserve the annual totals if requested. The procedure is an in-house SAS procedure built at Statistics Canada that is part of the Forillon project.

REFERENCES

1. Fortier, S. and B. Quenneville. 2009. *Reconciliation and Balancing of Accounts and Time Series*. JSM Proceedings, Business and Economic Section. Alexandria, VA: American Statistical Association.
2. Quenneville, B. and S. Fortier. 2006. *Balancing seasonally adjusted series as a complement to the direct and indirect approaches to seasonal adjustment*. JSM Proceedings, Business and Economic Section. Alexandria, VA: American Statistical Association.
3. SAS Institute Inc. 1991. *SAS/TOOLKIT® Software: Usage and Reference, Version 6, First Edition*. Cary, NC: SAS Institute Inc.
4. Latendresse, E., M. Djoura and S. Fortier. 2007. *Benchmarking Sub-Annual Series to Annual Totals – From Concepts to SAS® Procedure and SAS® Enterprise Guide® Custom Task*. Proceedings of the SAS Global Forum.
5. Dagum, E. B. and P. Cholette. 2006. *Benchmarking, Temporal Distribution and Reconciliation Methods of Time Series*. Springer-Verlag, New York, Lecture notes in Statistics, #186.
6. Schott, J. R. 2005. *Matrix Analysis for Statistic – 2nd edition*. John Wiley & Sons, Inc., New Jersey, Wiley Series in Probability and Statistics.

CONTACT INFORMATION

Your comments and questions are valued and encouraged, and should be sent to forillon@statcan.gc.ca. You may also contact the authors at:

Joana Bérubé
Statistics Canada
150 Tunney's Pasture Driveway
R.H. Coats, 17th floor
Ottawa, Ontario
K1A 0T6
Canada

(613) 951-6969
joana.berube@statcan.gc.ca

Susie Fortier
Statistics Canada
150 Tunney's Pasture Driveway
R.H. Coats, 17th floor
Ottawa, Ontario
K1A 0T6
Canada

(613) 951-4751
susie.fortier@statcan.gc.ca

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.